
Subject: Re: contour plots on map with Coyote Graphics System in postscript
Posted by [David Fanning](#) on Thu, 04 Oct 2012 22:33:00 GMT
[View Forum Message](#) <> [Reply to Message](#)

sasha.mozgova@gmail.com writes:

> Do you by any chance see what am I doing wrong?

I would do it like this:

```
PS_START, filename = work_path+'profiles_correlation_on_map.ps', $  
  /LANDSCAPE, /EUROPEAN
```

```
levels = cgConLevels(Float(data), NLevels=nlevels+1, $  
  MinValue=Floor(Min(data)), STEP=step, Factor=1)  
LoadCT, 0  
cgLoadCT, 2, /Reverse, /Brewer, NColors=nlevels-1, Bottom=1  
TVLCT, cgColor('white', /Triple), nlevels
```

```
map = Obj_New('cgMap', 'Equirectangular', Ellipsoid=19, $  
  XRange=xrange, YRange=yrange, /LatLon_Ranges, $  
  CENTER_LON=center_lon, $  
  Position=[0.1, 0.1, 0.9, 0.8], $  
  Limit=[Min(lat), Min(lon), Max(lat), Max(lon)])  
map -> Draw
```

```
cgContour, data, lon, lat, /Cell_Fill, /Overplot, /Outline, $  
  C_Colors=Indgen(nlevels)+1, Levels=levels, $  
  Map=map, OutColor='White'
```

```
cgMap_Grid, Map=map, /Box  
cgMap_Continents, Map=map, Color='tomato'  
cgColorbar, Range=[Min(levels), Max(levels)-step], OOB_High=nlevels, $  
  NColors=nlevels-1, Bottom=1, /Discrete, $  
  Title='Temperature $deg$) + 'K', $  
  Position=[0.1, 0.91, 0.9, 0.95]
```

```
PS_END
```

Cheers,

David

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Sepore ma de ni thue. ("Perhaps thos speakest truth.")
